Practical Probability with Spreadsheets

Chapter 6: DECISION VARIABLES

Case: SCOTIA SNOWBOARDS

The Scotia Corporation will sell snowboards next winter. Scotia's cost of manufacturing is $20 per snowboard. Scotia will get $48 in revenue for every snowboard that it sells next winter. If demand is less than supply, then the snowboards that Scotia still has unsold in its inventory at the end of the winter will have a value of $8 each to Scotia. If demand is greater than Scotia's supply, then excess demand will be lost to other competitors.

Scotia's factory and shipping schedules require that all snowboards that it sells next winter must be manufactured by September of this year, some three months before the winter season begins. As this year's manufacturing begins, Scotia does not have any old snowboards in inventory.

The predictions about demand for snowboards next winter depend on the general weather patterns, which may be normal or cold. If next winter's weather is normal, then demand for Scotia's snowboards will have probability 1/4 of being below 60,000, probability 1/2 of being below 75,000, and probability 3/4 of being below 90,000. On the other hand, if next winter's weather is cold then demand for Scotia's snowboards have probability 1/4 of being below 80,000, probability 1/2 of being below 100,000, and probability 3/4 of being below 125,000. (In either case, we may assume a Generalized Lognormal probability distribution.) It is currently estimated that the probability of cold weather next winter is 1/3, and the probability of normal weather is 2/3.

The staff engineer assigned to analyze this production planning decision has generally assumed that Scotia's objective in such situations should be to maximize the expected value of its profits. To verify this assumption, he asked Scotia's chief executive officer to think about a simpler gamble where Scotia could earn either $2 million or $1 million, each with probability 1/2, and asked whether Scotia would exchange this gamble for anything less than $1.5 million. The CEO replied, "Well, not for anything less than $1.4 million."

In a production planning session, a marketing manager remarked that good forecasts of the coming winter's general weather would be available in November. But the manufacturing director replied that a delay of snowboard production until November could substantially increase Scotia's total productions costs, perhaps by $100,000 or more.

1. Introduction

A decision variable is any quantity that we have to choose in a decision problem. In the Scotia case, the principal decision variable is the quantity of snowboards that Scotia will produce for next winter. Scotia's objective is to maximize its profit, which depends both on this production quantity and on the unknown demand for Scotia's snowboards. The demand for Scotia's snowboards next winter is the principal unknown quantity or random variable in the this case.

People sometimes confuse decision variables with random variables, because in some intuitive sense they are both "unknown" when we start a decision analysis. But of course we control a decision variable, and so we can stop being uncertain about it any time we want to make the decision. Random variables are used in decision-analysis models only to represent quantities that we do not know and cannot control.

To analyze a decision problem like the Scotia case, we may begin by constructing a simulation model that describes how profit returns and other outcomes of interest may depend on the decision variables that we control and on the random variables that we do not control. In this model, we can try specifying different values for the decision variables, and we can generate simulation tables that show how the probability distribution of profit outcomes could depend on the values of the decision variables. Using the criterion of expected profit maximization or some other optimality criterion, we can then try to find values of the decision variables that yield the best possible outcome distribution, where "best" may be defined by some optimality criterion like expected value maximization.

In this chapter, we will study some general techniques for analyzing decision problems with simulation, using the Scotia Snowboards case as our basic example. In Section 2, we consider three different ways that simulation can used to compare different decisions. In Section 3, we go beyond the expected value criterion and we introduce optimality criteria for decision-making with constant risk tolerance. In Section 4, we consider the strategic use of information.

2. General techniques for using simulation in decision analysis

Simulation is used in decision analysis to compare the probability distributions of payoff outcomes that would be generated by different decisions or strategies in a decision problem. In such simulation analysis, three different general approaches may be distinguished:

1. Make a separate simulation table of payoffs for each strategy. We can set up our simulation model assuming one strategy for the decision-maker and generate a large table of simulated payoffs for this strategy. Then we can make revised versions of the simulation model that assume another feasible strategies, one at a time, and similarly generate a separate table of simulated payoffs for each strategy that we want to consider. The difficulty with this approach is that our alternative strategies are being evaluated with different simulated values of the random variables. So a strategy that is always worse than some other strategies might appear better in such analysis, if the random variables that were used to generate its simulated payoffs happened to be more favorable (i.e., included more high-demand outcomes and fewer low-demand outcomes than the random data that were used to evaluate other strategies). Using larger data sets can reduce the probability of this kind of error, but it may be better to avoid it altogether by using the same simulated random variables to evaluate all of our alternative strategies.

2. Simultaneously simulate payoffs from alternative strategies. We can compute payoffs under several alternative strategies in our simulation model, with the same values of the simulated random variables, and the payoffs from these alternative strategies can be listed in the model output row of the simulation table. Then payoffs from all these strategies will be tabulated at once. So we will get a simulation table that has a payoff series for each strategy in one simulation table, where all these payoff series were computed using the same series of simulated random variables. The difficulty with this method is that it requires us to specify in advance the strategies that we want to evaluate, and the simulation table can become big and unwieldy if we want to evaluate many strategies.

3. Tabulate simulated values of all payoff-relevant random variables. If the number of random variables that influence payoffs under all strategies is not too large, then we can make a table of simulated values of these random variables. With this fixed simulation table of all payoff-relevant random variables, we can then compute payoffs for as many different strategies as we like, and the expected payoffs and other summary statistics can be summarized for different decisions in a data table. With this approach, we can even ask Excel Solver to search for strategies that maximize the expected payoff or some other target formula. This approach can become unwieldy and infeasible, however, if the number of random variables that we need to tabulate is very large, in which case in which case we should use approach 2 above instead.

Let us illustrate these ideas in the context of the Scotia Snowboards case. To begin, consider Figure 1, which applies the first approach described above, that of generating a separate simulation table of payoffs for each possible decision option.

The basic parameters of the Scotia case (demand quartiles in each weather state, probability of cold weather, prices and costs) are summarized in the range A1:C12. The random weather pattern (with 0 for "normal" and 1 for "cold") is simulated in cell B15. Cells E16:E19 contain IF formulas that return the quartile boundaries for the conditional distribution of demand given the simulated weather pattern in B15. Then demand for Scotia snowboards is simulated in cell B16, with Generalized Lognormal random variable that has the quartiles listed in E16:E19. So the probability distribution of simulated demand in cell B16 depends on the simulated weather pattern in cell B15 just as Scotia's demand is supposed to depend on the actual weather pattern.

Cell C17 in Figure 1 is used to represent the quantity of snowboards that Scotia will produce, which is the principal decision variable in this model. So the current value of cell C17 in Figure 1, which is 85,000, should be interpreted as just one tentative proposal, to be evaluated in the model. Scotia's actual sales of snowboards will either the demand quantity or the production quantity, whichever is smaller. So cell B18, with the formula =MIN(B16,C17) contains the sales quantity (in snowboards) that would result from the production quantity in C17 and the demand in B16. Then the remaining inventory of snowboards at the end of the season (which have a remainder value of $8 per snowboard) can be computed by the formula =C17−B18 in cell B19, and the final profit (in dollars) can be computed by the formula

=B11*B18+B12*B19−B10*C17

in cell B20. (Cell B10 contains the cost per snowboard, B11 contains the selling price, and B12 contains the value of snowboards in inventory at the end of the winter.)

	A	B	C	D	E	F	G	H
1	SCOTIA PARAMETERS:				FORMULAS			
2	Demand distribution given weather				B15. =IF(RAND()<B8,1,0)			
3	Quartiles	Normal	Cold		E16. =IF($B$15=1,C4,B4)			
4	q1 (.25)	60000	80000		E16 copied to E16:E18			
5	q2 (.50)	75000	100000		B16. =GENLINV(RAND(),E16,E17,E18)			
6	q2 (.75)	90000	125000		B18. =MIN(B16,C17)			
7					B19. =C17-B18			
8	P(cold winter)	0.33333			B20. =B11*B18+B12*B19-B10*C17			
9					B28. =B20			
10	$Cost/unit	20			B23. =AVERAGE(B29:B429)			
11	$SellingPrice	48			B24. =STDEV(B29:B429)			
12	RemainderValue	8			B25. =PERCENTILE(B29:B429,0.1)			
13					B26. =PERCENTILE(B29:B429,0.9)			
14	Simulation Model							
15	Weather cold?	0			Demand quartiles | weather			
16	Demand	93888			60000			
17	Production Quantity		85000		75000			
18	Sales	85000			90000			
19	Remainder	0						
20	Profit ($)	2380000						
21								
22	Profit statistics:							
23	E(Profit)	1934476						
24	Stdev	622816						
25	10%-level	987162						
26	90%-level	2380000						
27		Sim'd Profit						
28	SimTable	2380000						
29	0	-960027						
30	0.0025	-390261						
31	0.005	-125345						
32	0.0075	13019						
33	0.01	22501.1						
34	0.0125	84215.5						
35	0.015	86416.1						
36	0.0175	124765						
37	0.02	195970						
38	0.0225	256098						
39	0.025	329552						
40	0.0275	362521						
41	0.03	405190						
42	0.0325	453879						
43	0.035	525338						
44	0.0375	532715						



Figure 1.

Cell B28 in Figure 1 echoes the simulated profit in cell B20 (by the formula =B20), and 401 simulated values of this simulated profit have been tabulated below in cells B29:B429. The profit data has been sorted to make the cumulative risk profile shown in the Figure 1, which plots the percentile index in cells A29:A429 on the horizontal axis and the simulated profit values in cells B29:B429 on the vertical axis. (Because of this sorting, Figure 1 shows only the worst profit outcomes at the top of the table in cells B29:B44.) The expected profit for Scotia with the production quantity of 85,000 snowboards is estimated to be approximately $1,934,000 in cell B23 by the formula =AVERAGE(B29:B429),

Figure 1 only shows us an analysis of one possible decision: that of producing 85,000 snowboards. But it is easy for us to enter other possible values of this decision variable into cell C17, and regenerate the simulation table. By entering the production quantities 80,000 through 85,000 successively into cell C17 and regenerating the simulation table in this spreadsheet once for each production quantity, I got the estimates of expected profit in cell B23 as follows

Production quantity	Average simulated profit
80,000	1,930,656
81,000	1,891,684
82,000	1,909,517
83,000	1,851,352
84,000	1,953,704
85,000	1,934,476

As we move down this table, we find that small increases in the production quantity are estimated to have rather large changes in expected profit which alternate strangely between increases and decreases. Most striking is the result of increasing production from 83,000 to 84,000, which is estimated here to increase expected profit by more than $100,000, even though we cannot earn more than $28 per snowboard! This absurd conclusion is a result our using different simulated demands to evaluate different production quantities. In my simulations, large demands may have occurred more frequently in the simulations when I was evaluating the production quantity 84,000 in cell C17, and less frequently in the simulations when I was evaluating the production quantity 83,000 in cell C17.

To say which production quantity is better under the expected value criterion, we need to estimate how expected profit changes when the production quantity changes. That is, what we really need to estimate accurately are the differences between expected profits under different production quantities. These differences can be estimated are more accurately in problems like Scotia when the various production quantities are evaluated with the same series of simulated demands, as illustrated in Figure 2.

Method 2 (as described in the beginning of this section) is applied in Figure 2 for the selected production quantities of 85,000, 95,000, 105,000, and 115,000, which have been entered into cells B20:E20. The simulated demands to compare these alternative decisions are generated by the random variables in cells B16 and C16 of Figure 2. Cell B16 simulates the weather pattern by the formula

=IF(RAND()<B8,1,0),

where B8 contains the value 1/3 = P(Cold). The simulated demand in cell C16 is generated by the formula

=IF(B16=1, GENLINV(RAND(),C4,C5,C6), GENLINV(RAND(),B4,B5,B6)

where cells C4:C6 contain the demand quartiles for cold weather, and cells B4:B6 contain the demand quartiles for normal weather.

The profit that would result from the demand in cell C16 and the production quantity in cell B19 is computed in cell B29 of Figure 2 by the formula

=($B$12−$B$10)*B$20+($B$11-$B$12)*MIN(B$20,$C16)

To interpret this formula, notice that B12 contains the remainder value and B10 contains the cost per unit, so (B12−B10)*B20 is the (negative) profit that would result from producing the quantity B20 and leaving all production in our final inventory. But then taking each unit out of inventory and selling it at the price B11 increases profit by B11−B12, and the number of units sold is the minimum of production and demand MIN(B20,C16). Copying this formula from B29 to B29:E29 gives us the alternative profits that would result from the alternative production quantities in B20:E20, all with the same simulated demand in C16.

	A	B	C	D	E	F	G	H	I	J	K	L
1	**SCOTIA SNOWBOARDS PARAMETERS:**			**FORMULAS**								
2	**Demand distribution |weather**			**B16. =IF(RAND()<B8,1,0)**								
3	**Quartiles**	**Normal**	**Cold**	**C16. =IF(B16=1,GENLINV(RAND(),C4,C5,C6),GENLINV(RAND(),B4,B5,B6))**								
4	**q1 (.25)**	**60000**	**80000**	**B29. =($B$12-$B$10)*B$20+($B$11-$B$12)*MIN(B$20,$C16)**								
5	**q2 (.50)**	**75000**	**100000**	**B29 copied to B29:E29**								
6	**q3 (.75)**	**90000**	**125000**	**B22. =AVERAGE(B30:B430)**								
7				**B23. =STDEV(B30:B430)**								
8	**P(cold winter)**	**0.33333**		**B24. =PERCENTILE(B30:B430,0.1)**								
9				**B25. =PERCENTILE(B30:B430,0.9)**								
10	**$Cost/unit**	**20**		**B22:B25 copied to B22:E26**								
11	**$SellingPrice**	**48**		B27. =RISKTOL(2000000,1000000,1400000)								
12	**RemainderValue**	**8**		B26. =CE(B30:B430,$B$27)								
13				B26 copied to B26:E26								
14		**SIMULATION MODEL**										
15		**Cold?**	**Demand**									
16		**0**	**75396**									
17												
18	**ANALYSIS OF FOUR ALTERNATIVE DECISIONS**											
19		**With Production Quantity =**										
20		**85000**	**95000**	**105000**	**115000**							
21	**Profit Statistics:**											
22	**E(Profit)**	**1940908**	**1968541**	**1949524**	**1895968**							
23	**Stdev**	**587875**	**713670**	**822683**	**910784**							
24	**10%-level**	**1066015**	**946015**	**826015**	**706015**							
25	**90%-level**	**2380000**	**2660000**	**2940000**	**3220000**							
26	CertainEquiv	1758032	1716720	1637468	1537206							
27	with RiskTol	1216303										
28	**Sim'd profits with ProdnQ as above**											
29	**SimTable**	**1995827**	**1875827**	**1755827**	**1635827**							
30	**0**	**1339507**	**1219507**	**1099507**	**979507**							
31	**0.0025**	**259480**	**139480**	**19480**	**-100520**							
32	**0.005**	**2380000**	**2660000**	**2940000**	**3220000**							



Figure 2.

The range B30:E430 in Figure 2 contains a table of profit outcomes for these four production quantities with 401 independent simulations of the demand random variable B16. The sample averages of these simulated profits are computed in cells B22:E22 to estimate the expected profit for each production quantity. By the criterion of expected profit maximization, the best of these four production quantities is apparently 95,000, which yields an estimated expected profit of $1,968,541. These estimated expected profits are plotted above their corresponding production quantities by the thick curve in the chart in Figure 2.

The top and bottom curves in the chart in Figure 2 show the values at the 0.90 and 0.10 cumulative probability levels in the profit distributions that result from these four alternative production quantities (as computed in cells B24:E25). The increasing spread between these curves shows that higher production quantities may also be associated with more risk. (The certainty equivalents which are computed in B26:E26 of Figure 2 and are plotted by the dotted line in the chart will be discussed later in Section 3.)

Figure 3 applies the third approach to the Scotia case: tabulating the random variables first and then computing profits from this fixed simulation data by formulas that depend "live" on a cell that represents the decision variable. The random variables for weather and demand are simulated in cells B22 and C22 of Figure 3 by the formulas

```
=IF(RAND()<B8,1,0)
```

to simulate the weather pattern (cold=1, normal=0) in cell B22, and

```
=IF(B22=1,GENLINV(RAND(),C4,C5,C6),GENLINV(RAND(),B4,B4,B6)
```

to simulate Scotia's snowboard demand in cell C22. Fixed data from 1001 simulations of these random variables has been stored below in the range B23:C1023. So the expected value of demand can be estimated to be 85,016, according to the average of sampled demands which is computed in cell C18.

	A	B	C	D	E	F	G	H	I	J	K
1	SCOTIA SNOWBOARDS PARAMETERS:			FORMULAS FROM RANGE A1:E1023							
2	Demand distribution |weather			B22. =IF(RAND()<B8,1,0)							
3	Quartiles	Normal	Cold	C22. =IF(B22=1,GENLINV(RAND(),C4,C5,C6),GENLINV(RAND(),B4,B5,B6))							
4	q1 (.25)	60000	80000	C18. =AVERAGE(C$23:C$1023)							
5	q2 (.50)	75000	100000	C19. =STDEV(C$23:C$1023)							
6	q3 (.75)	90000	125000	C18:C19 copied to E18:E19							
7				E23. =($B$12-$B$10)*$D$18+($B$11-$B$12)*MIN($D$18,C23)							
8	P(Cold)	0.33333		E23 copied to E23:E1023							
9				SOLVER: maximize E18 by changing D18.							
10	$Cost/unit	20		B15. =RISKTOL(2000000,1000000,1400000)							
11	$SellingPrice	48		E20. =CE(E23:E1023,$B$15)							
12	RemainderVal	8		SOLVER: maximize E20 by changing D18.							
13											
14	For CEs:										
15	RiskTolerance	1216303		DECISION ANALYSIS				E(Profit)	10%ile	90%ile	CE
16				Without forecast			(D18)	1949540	887409	2673132	1685872
17			Demand	ProdnQ	Profit statistics		70000	1759590	1193037	1960000	1673159
18		Mean	85016	95469	1949540	E(Profit)	80000	1883216	1073037	2240000	1730584
19		Stdev	31020		740061	Stdev	90000	1942678	953037	2520000	1716623
20	SIMULATION MODEL				1685872	CE	100000	1944749	833037	2800000	1652179
21		Cold?	Demand				110000	1903259	713037	3080000	1558832
22	SimTable	0	87378		Profit		120000	1836918	593037	3360000	1452116
23	0.000	0	50741		884020						
24	0.001	1	82064		2136931		FORMULAS FROM RANGE G15:K22				
25	0.002	1	83448		2192276		H16. =E18		K16. =E20		
26	0.003	0	92085		2537785		I16. =PERCENTILE(E23:E1023,0.1)				
27	0.004	0	39506		434593		J16. =PERCENTILE(E23:E1023,0.9)				
28	0.005	0	72384		1749713		H17:K22. {=TABLE(,D18)}				
29	0.006	0	85538		2275875						
30	0.007	0	63936		1411823						

Figure 3.

A possible production quantity has been entered into cell D19 of Figure 3, and the profits that would result from this production quantity for each of the simulated demand values are computed in E23:E1023 by entering the formula

=($B$12−$B$10)*$D$18+($B$11-$B$12)*MIN($D$18,C23)

into cell E23, and then copying E23 to cells E23:E1023. Then the expected profit with the production quantity D18 is computed in cell E18 by the formula

=AVERAGE(E23:E1023)

The big advantage of this approach is that this estimated expected profit in cell E18 of Figure 3 depends actively on the decision variable in cell D18. We can try different decisions in D18 and watch how the expected profits change in E18. A data table can be used to show how expected profit and other statistics of the simulated profit distribution would vary with any group of selected production quantities, as illustrated in G15:K22 of Figure 3.

We can even ask Solver to search for the value of our decision variable in cell D18 that would maximize the expected profit in cell E18. The value 95,469 that is shown in cell D18 in Figure 3 is actually the result of Solver's optimization, when directed to maximize E18 by changing D18. (The global optimality of this result was confirmed by repeated reapplications of Solver with different initial conditions, some much higher and others much lower, all of which terminated with this production quantity reported as the optimal solution.) So we may estimate that 95,000 snowboards is Scotia's optimal production quantity, according to the criterion of expected profit maximization, and that Scotia can achieve an expected profit of about $1,950,000 with this production quantity. (A similar conclusion might have been taken from the analysis in Figure 2 with different simulation data, except that there we had no way to check whether higher expected profits could have been achieved by other production quantities between 85,000 and 95,000, or between 95,000 and 105,000.)

Notice that this optimal production quantity for maximizing the expected value of profit is quite different from the expected value of demand which we estimated in cell C18 of Figure 3. If demand were perfectly predictable then the profit-maximizing production quantity would be equal to demand. But when there is uncertainty about demand, the optimal production quantity is not necessarily equal to expected demand, even when our goal is to maximize expected profit.

3. Taking account of risk aversion

In the decision analysis literature, a decision-maker is called risk-neutral if he (or she) is willing to base his decisions purely on the criterion of maximizing the expected value of his monetary income. The criterion of maximizing expected monetary value is so simple to work with that it is often used as a convenient guide to decision-making even by people who are not perfectly risk neutral. But in some situations, people often feel that comparing gambles only on the basis of expected monetary values would take insufficient account of their aversion to risks.

For example, imagine that you had a lottery ticket that would pay you either $20,000 or $0, each with probability 1/2. If you are risk neutral, then you should be unwilling to sell this ticket for any amount of money less than its expected value, which is $10,000. But many risk averse people might be very glad to exchange this risky lottery for a certain payment of $9000.

Given any such lottery or gamble that promises to pay you an amount of money that will be drawn randomly from some probability distribution, a decision-maker's certainty equivalent of this gamble is the lowest amount of money-for-certain that the decision-maker would be willing to accept instead of a gamble. That is, if $7000 would be your certainty equivalent of the lottery that would pay you either $20,000 or $0, each with probability 1/2, then you should be just indifferent between having a ticket to this lottery or having $7000 cash in hand.

In these terms, a risk-neutral person is one whose certainty equivalent of any lottery is just equal to its expected monetary value. A person is risk averse if his or her certainty equivalent of a gamble is less than the gamble's expected monetary value. The difference between the expected monetary value of a gamble and a risk-averse decision-maker's certainty equivalent of the gamble is called the decision-maker's risk premium for the gamble. (RP=EMV−CE.)

When you have a choice among various gambles, you should choose the one for which you have the highest certainty equivalent, because it is the one that you would value the most highly. But when a gamble is complicated, you may find it difficult to assess your certainty equivalent for it. The great appeal of the risk-neutrality assumption is that, by identifying your certainty equivalent with the expected monetary value, it makes your certainty equivalent something that is straightforward to compute or estimate by simulation. So what we need now is to find more general formulas that risk-averse decision-makers can use to compute their certainty equivalents for complex gambles and monetary risks.

A realistic way of calculating certainty equivalents must include some way of taking account of a decision-maker's personal willingness to take risks. The full diversity of formulas that a rational decision-maker might use to calculate certainty equivalents is described by a branch of economics called utility theory. Utility theory generalizes the principle of expected value maximization in a simple but very versatile way. Instead of assuming that people want to maximize their expected monetary values, utility theory instead assumes that each individual has a personal utility function that assigns a utility values to all possible monetary income levels that the individual might receive, such that the individual always wants to maximize his or her expected utility value. So according to utility theory, if the random variable **X** denotes the amount of money that some risky gamble would pay you, and if U(y) denotes your utility for any amount of money y, then your certainty equivalent CE of this gamble should be the amount of money that gives you the same utility as the expected utility of the gamble. That is, we have the basic equation

$$U(CE) = E(U(\mathbf{X})).$$

Utility theory can account for risk aversion, but it also is consistent with risk neutrality or even risk-seeking behavior, depending on the shape of the utility function. In 1947, von Neumann and Morgenstern gave an ingenious argument to show that any consistent rational decision maker should choose among risky gambles according to utility theory. Since then, decision analysts have developed techniques to assess individuals' utility functions. Such assessment can be difficult, because people have difficulty thinking about decisions under uncertainty, and because there as so many possible utility functions. But just as we could simplify subjective probability assessment by assuming that an individual's subjective probability distribution for some unknown quantity might be in some natural mathematical family of probability distributions (like the Generalized Lognormal family), so we can simplify the process of subjective utility assessment by assuming that a decision-maker's utility function in some natural family of utility functions.

For practical decision analysis, the most convenient utility functions to use are those that have a special property called constant risk tolerance. Constant risk tolerance means that, if we change a gamble by adding a fixed additional amount of money, about which there is no uncertainty, then the certainty equivalent should increase by the same amount. This assumption of constant risk tolerance gives us a simple one-parameter family of utility functions: If a decision-maker is are risk averse and has constant risk tolerance, then the decision-maker's preferences can be described by a utility function of the

$$U(x) = -EXP(-x/r)$$

where r is a number that is the decision-maker's risk tolerance constant.

This assumption of constant risk tolerance is very convenient in practical decision analysis, because it allows us to evaluate independent gambles separately. If you have constant risk tolerance then, when you are going to earn money from two independent gambles, your certainty equivalent for the sum of the two independent gambles is just the sum of your certainty equivalent for each gamble by itself. That is, your certainty equivalent value of a gamble is not affected by having other independent gambles, provided that you have one of these constant-risk-tolerance utility functions.

So in this course, we will focus on decision analysis where the decision-maker is assumed to have a utility function with constant risk tolerance. With this assumption, we can take some account of an individual's personal attitudes towards risk and yet still have a method for estimating certainty equivalents that is straightforward to apply in realistic situations.

Simtools provides five functions to simplify such risk analysis: CE, RISKTOL, UTIL, UINV, and CEPR. Of these functions, CE and RISKTOL will be the most useful. So let us begin by explaining how to use them.

Once we have admitted that some people have less tolerance for risk than others, we need some way to measure an individual's risk tolerance. The way to do this is to ask the individual to think about some relatively simple gamble and assess his or her personal certainty equivalent for this gamble. From that subjectively assessed certainty equivalent, we can mathematically compute the individual's risk tolerance, which we can then use to compute the individual's certainty equivalent for much more complicated gambles.

The simplest gambles to think about are gambles that involve just two equally likely outcomes, each with probability 1/2. So the RISKTOL function is designed to compute the risk tolerance from subjectively assessed certainty equivalents for such simple binary lotteries.

RISKTOL takes three parameters, each of which must be a number. RISKTOL's first parameter may be called HighIncome or H, its second parameter may be called LowIncome or L, and its third parameter may be called CertainEquiv or C. Then the formula RISKTOL(H,L,C) returns the risk tolerance constant that for a decision-maker for whom C is the certainty equivalent of a simple lottery that would pay either the high monetary income H or the low monetary income L, each with probability 1/2. The numbers H, L, and C here must all be measured in the same monetary units (say, dollars) and must satisfy the inequalities H > C > L. The risk tolerance constant is measured in the same monetary units.

So if you think that a lottery ticket paying either $20,000 or $0, each with probability 1/2, would be just worth $7000 to you, then (assuming that you have constant risk tolerance) your risk tolerance must be RISKTOL(20000, 0, 7000) = $15,641. This measure of risk tolerance may be quite difficult to interpret. The important thing to recognize is that lower levels of risk tolerance correspond to certainty equivalents that are farther below the expected monetary value. A more risk averse individual whose certainty equivalent for this same lottery ticket was only $5000 would have a lower risk tolerance RISKTOL(20000, 0, 5000) = $8205. A more risk tolerant individual whose certainty equivalent for this same lottery ticket was $9900 would have a much higher risk tolerance RISKTOL(20000, 0, 9500) = $99,833. A risk-neutral individual, whose certainty equivalent for this lottery would be the expected monetary value $10,000, has an infinite risk tolerance, which RISKTOL(20000, 0, 10000) represents with a "dividing by zero" error message.

In practice, because the assumption of constant risk tolerance is not a perfect fit for most people's preferences, it is best to assess risk tolerances by using hypothetical binary lotteries where the levels and spread of the two prizes are similar to the levels and spread of the possible outcomes in the actual decision situation that we are studying. So in the Scotia case, where likely monetary incomes can vary from $1 million to $3 million (recall the chart in Figure 2), it seems appropriate to try to quantify top management's willingness to take risks in this situation by asking the chief executive officer to think about a hypothetical binary lottery that would pay either $1 million or $2 million, each with probability 1/2. Given his response, indicating $1.4 million as the certainty equivalent for this lottery, we may infer that Scotia management may want to analyze this decision problem using a risk tolerance constant

RISKTOL(2, 1, 1.4) = $1.216 million.

Notice that if we do the accounting in dollars instead of $millions, then we get the equivalent answer, RISKTOL(2000000, 1000000, 1400000) = 1,216,303, as shown in cell D23 of Figure 4.

The CE function has two required parameters. The first parameter of CE must be a spreadsheet range that contains monetary income values. The second required parameter of CE must be a number r that represents the decision-maker's risk tolerance. Then the formula CE(incomes, r) returns the certainty equivalent of a random draw from the values in the incomes range (each with equal probability) for a decision-maker who has the risk tolerance constant R.

From the equation RISKTOL(20000,0,7000) = 15641, we saw that a decision-maker with risk tolerance constant $15641 would have certainty equivalent $7000 for a lottery that would be equally likely to pay $20000 or $0. So if you enter the value 20000 into cell A1 and enter the value 0 into cell A2, then you should find that the formula =CE(A1:A2,15641) returns the value 7000.

An application of the CE function to the Scotia case is shown in Figure 4. The range B29:B429 contains a sample of 401 simulated profit values for Scotia when its production quantity is 85,000 snowboards. Cell D23 contains the risk tolerance constant that we computed from the chief executive officer's assessment that a simple binary lottery between $2 million and $1 million would be worth $1.4 million to Scotia. So the formula

=CE(B29:B429,D23)

in cell D24 of Figure 4 returns the certainty equivalent value of $1,727,635 for a random draw from these 401 simulated profit values. Because these 401 values in B29:B429 can be taken as a good approximation to the probability distribution of profits out of which they have been sampled, we may take $1,728,000 is our estimate of Scotia's certainty equivalent of the gamble that it would get by producing 85,000 snowboards this year. Notice that this certainty equivalent is somewhat less than the expected profit value of $1,934,476 computed in cell B23, because of Scotia's risk aversion. (So Scotia's risk premium for this gamble is the difference RP = EMV−CE = $206,841.)

	A	B	C	D	E	F	G	H
1	SCOTIA PARAMETERS:				FORMULAS			
2	Demand distribution given weather				B15. =IF(RAND()<B8,1,0)			
3	Quartiles	Normal	Cold		E16. =IF($B$15=1,C4,B4)			
4	q1 (.25)	60000	80000		E16 copied to E16:E18			
5	q2 (.50)	75000	100000		B16. =GENLINV(RAND(),E16,E17,E18)			
6	q2 (.75)	90000	125000		B18. =MIN(B16,C17)			
7					B19. =C17-B18			
8	P(cold winter)	0.33333			B20. =B11*B18+B12*B19-B10*C17			
9					B28. =B20			
10	$Cost/unit	20			B23. =AVERAGE(B29:B429)			
11	$SellingPrice	48			B24. =STDEV(B29:B429)			
12	RemainderValue	8			B25. =PERCENTILE(B29:B429,0.1)			
13					B26. =PERCENTILE(B29:B429,0.9)			
14	Simulation Model							
15	Weather cold?	0			Demand quartiles | weather			
16	Demand	62838			60000			
17	Production Quantity		85000		75000			
18	Sales	62838			90000			
19	Remainder	22162						
20	Profit ($)	1493503						
21								
22	Profit statistics:					Hi $	Low $	CE
23	E(Profit)	1934476		1216303	RiskTol	2000000	1000000	1400000
24	Stdev	622816		1727635	CE			
25	10%-level	987162		-0.2416	EU			
26	90%-level	2380000		1727635	UINV(EU)	401	Sample Size	
27		Sim'd Profit				0.00986	Stdev(EU)	
28	SimTable	1493503		Utility($)	95% confidence interval			
29	0	2380000		-0.1413	-0.2609	-0.2223	EU	
30	0.0025	2365707		-0.143	1634031	1829048	CE	
31	0.005	2380000		-0.1413				
32	0.0075	1481498		-0.2958	FORMULAS FROM RANGE D23:G30			
33	0.01	2026114		-0.189	D23. =RISKTOL(F23,G23,H23)			
34	0.0125	2078718		-0.181	D24. =CE(B29:B429,D23)			
35	0.015	1454374		-0.3025	D29. =UTIL(B29,$D$23)			
36	0.0175	1440798		-0.3059	C29 copied to D29:D429			
37	0.02	922506		-0.4684	D25. =AVERAGE(D29:D429)			
38	0.0225	1029177		-0.4291	D26. =UINV(D25,$D$23)			
39	0.025	2380000		-0.1413	F26. =COUNT(D29:D429)			
40	0.0275	2380000		-0.1413	F27. =STDEV(D29:D429)/(F26^0.5)			
41	0.03	1822914		-0.2234	E29. =D25-1.96*F27			
42	0.0325	2380000		-0.1413	F29. =D25+1.96*F27			
43	0.035	2380000		-0.1413	E30. =UINV(E29,$D$23)			
44	0.0375	1638680		-0.26	F30. =UINV(F29,$D$23)			

Figure 4.

Figure 4 also illustrates how the CE function is actually computed. As we have seen, a decision-maker with constant risk tolerance R has a utility function for money of the form U(x) = −EXP(−x/r). To make this formula easier to remember, Simtools gives us a function UTIL(x,r) that does the same thing. That is,

UTIL(x,r) = −EXP(−x/r)

Simtools also provides an inverse function UINV that converts expected utility values back to the equivalent monetary amount. That is, for any possible expected utility value u, the formula UINV(u,r) is the monetary amount that would have utility u when the risk tolerance constant is r. This Simtools formula UINV(u,r) is equivalent to the Excel formula −r*LN(−u). That is,

UINV(u,r) = −r*LN(−u)

So with risk tolerance constant in cell D23, the utility values of the simulated profits in Figure 4 are computed by entering the formula

=UTIL(B29,$D$23)

into cell D29, and then copying cell D29 to D29:D429. Then the expected utility from this distribution of profits is computed in cell D25 by the formula

=AVERAGE(D29:D429)

In cell D26 of Figure 4, the formula

=UINV(D25,D23)

returns the amount of money that would have utility value equal to cell D25, when D23 is the risk tolerance constant. (If you entered the formula =UTIL(D26,D23) in this spreadsheet, the result would be the expected utility value in cell D25.) Notice that the value returned by UINV in cell D26 is exactly the same as the value returned by CE in cell D24. So what the CE function converts the incomes values to utilities, computes the average of these utilities, and then converts this average utility back to return the certainty equivalent value of monetary income.

It is difficult to interpret the utility values in cells D29:D429 of Figure 4. The fact that they are negative has no significance at all. More important is the shape of the utility function, which is shown in Figure 5. Notice that the higher monetary income levels are associated with higher utility levels, but the rate of utility increase becomes slower as income increases. This shape of the utility function expresses that an extra dollar when your income is low would increase your utility by more than an extra dollar when your income is high. It is this greater sensitivity to income changes when poor that makes people risk averse.

	A	B	C	D	E	F	G
1	RiskTolerance						
2	1.2163						
3							
4							
5	$X	UTIL(X,RT)	UINV(U,RT)				
6	-1	-2.2754	-1				
7	-0.75	-1.8527	-0.75				
8	-0.5	-1.5084	-0.5				
9	-0.25	-1.2282	-0.25				
10	0	-1.0000	-0				
11	0.25	-0.8142	0.25				
12	0.5	-0.6629	0.5				
13	0.75	-0.5398	0.75				
14	1	-0.4395	1				
15	1.25	-0.3578	1.25				
16	1.5	-0.2913	1.5				
17	1.75	-0.2372	1.75				
18	2	-0.1931	2				
19	2.25	-0.1573	2.25				
20	2.5	-0.1280	2.5		FORMULAS		
21	2.75	-0.1043	2.75		A2. =RISKTOL(2,1,1.4)		
22	3	-0.0849	3		B6. =UTIL(A6,$A$2)		
23	3.25	-0.0691	3.25		C6. =UINV(B6,$A$2)		
24	3.5	-0.0563	3.5		B6:C6 copied down		
25	3.75	-0.0458	3.75				
26	4	-0.0373	4				



Figure 5.

The CE function is intended as a handy shortcut to let you avoid looking at these unintuitive utility numbers, because all that we generally care about is the final certainty equivalent value. But there is one computational task that may require us to go behind the CE function and look at the utility numbers: when we want to compute a 95% confidence interval for the true certainty equivalent that we are estimating based on a sample of simulated profit values. The 95% confidence formula for an unknown expected value cannot be applied directly to the certainty equivalent, because it is not an expected value, but this formula can be applied to the expected utility value. Given the profit data in B29:B429 and the corresponding utility values in D29:D429 in Figure 4, the average utility in cell D25 is our estimate of the unknown expected utility of the true profit distribution out of which our profit data was drawn. The standard deviation of this average utility can be estimated by the formula

=STDEV(D29:D429)/COUNT(D29:D429)^0.5

which is computed in cell F27. The endpoints of the 95% confidence interval for the unknown expected utility can then be computed by the formulas

=D25−1.96*F27 and =D25+1.96*F27

which have been entered into cells E29 and F29. With the risk tolerance constant D23, these low and high expected utility values correspond to monetary income levels that are returned by UINV in the formulas

=UINV(E29,$D$23) and =UINV(F29,$D$23)

as shown in cells E30 and F30 of Figure 4. These monetary values $1,634,031 and $1,829,048 are the endpoints of our 95% confidence interval for Scotia's true certainty equivalent of the probability distribution of profits that would be generated by the production quantity 85,000.

In Figure 2, the certainty equivalents of profit for the four selected production quantities are estimated in cells B26:E26. With Scotia's risk tolerance constant in cell B27 and the simulated profit values for each production quantity in rows 30 to 430, these estimated certainty equivalents are computed by entering the formula

=CE(B30:B430,$B$27)

into cell B26, and then copying cell B26 to cells B26:E26.

The highest of the four certainty in Figure 2 is in cell B29, generated by the production quantity 85,000. Thus, a decision analysis that takes Scotia's risk aversion into account may leads us to recommend a lower optimal production quantity than a decision analysis that assumes risk neutrality.

In cells B20 and C20 of Figure 2, we see that the expected profit from producing 85,000 appears only slightly lower than the expected profit from producing 95,000. But cells B24 and C24 show that the low-end profit level with 10% cumulative probability is substantially higher when production is 85,000 than when it is 95,000 units. Notice also that the certainty equivalent curve in the chart in Figure 2 is always below the expected monetary value (mean) curve, but the certainty equivalent is farther below the expected monetary value for the higher production quantities where the risks are greater (as indicated by the spread between 10% and 90% cumulative probability levels of profit). Thus, it is intuitively reasonable that taking risk aversion into account should lead Scotia to reduce its production quantity somewhat.

Now let us return to the spreadsheet shown in Figure 3, where the profits in E23:E1023 can be recalculated for any production quantity in cell D18. With Scotia's risk tolerance in cell B15, the formula

=CE(E23:E1023,D18)

in cell E20 gives us a certainty equivalent (in dollars) for the profit distribution generated by any production quantity in D18. Then we can ask Solver to maximize this certainty equivalent in cell E20 by changing the production quantity in cell D18. With the given simulated demand data in this spreadsheet, Solver found the CE-maximizing production quantity to be 82,524 snowboards, which yielded a certainty equivalent of $1,733,204. Using this optimal production quantity, the mean and standard deviation of Scotia's profits, as computed in cells C18 and C19 of Figure 3, were found to be $1,904,326 and $573,859 respectively. Thus, compared to the risk-neutral optimization shown in Figure 3 (where the production quantity is 95,469), we again find that risk aversion makes Scotia reduce its production somewhat, which decreases expected profit by about 2% but decreases the standard deviation of profit by about 20%.

(Notice that Figures 2, 3, and 4 may get different estimates of similar quantities, because these spreadsheets are using different independently-generated simulation data. In particular, cell B26 in Figure 2 and cell D24 in Figure 4 are both estimates of the same certainty equivalent of profit with production 85,000. The estimates differ by only about $30,000, and the value of cell B26 is well within the 95% confidence interval that we computed in Figure 4.)

	A	B	C	D	E	F	G	H	I	J	K	L
1	SCOTIA SNOWBOARDS PARAMETERS:			FORMULAS FROM RANGE A1:L1023								
2	Demand distribution |weather			B22. =IF(RAND()<B8,1,0)								
3	Quartiles	Normal	Cold	C22. =IF(B22=1,GENLINV(RAND(),C4,C5,C6),GENLINV(RAND(),B4,B5,B6))								
4	q1 (.25)	60000	80000	E23. =($B$12-$B$10)*$D$18+($B$11-$B$12)*MIN($D$18,C23)								
5	q2 (.50)	75000	100000	G23. =IF(B23=$G$21,E23,"..")								
6	q3 (.75)	90000	125000	J23. =VLOOKUP(B23,$I$18:$J$19,2,0)								
7				K23. =($B$12-$B$10)*J23+($B$11-$B$12)*MIN(J23,C23)								
8	P(Cold)	0.33333		E23:K23 copied to E23:K1023.								
9				E18. =AVERAGE(E23:E1023)				E18 copied to B18.				
10	$Cost/unit	20		E19. =STDEV(E23:E1023)								
11	$SellingPrice	48		E18:E19 copied to G18:G19,K18:K19.								
12	RemainderVal	8		K15. =K18-E18								
13				SOLVER: maximize K18 by changing J18:J19.								
14											E(ValueOfInfo)	
15				DECISION ANALYSIS							45102	
16				Without forecast			Conditional		Strategy with forecast			
17	P(Cold) from SimTable			ProdnQ			on weather		with:	ProdnQ	Profit	
18		0.33267		95469	1949540	E(Profit)	2317261		0	87226	1994642	E(Profit)
19					740061	Stdev	538988		1	118443	793175	Stdev
20	SIMULATION MODEL						with (Cold?)=					
21		Cold?	Demand				1					
22	SimTable	0	63434		Profit	Profit |	(Cold?)			ProdnQ	Profit	
23	0.000	0	50741		884020		..			87226	982936	
24	0.001	1	82064		2136931		2136931			118443	1861243	
25	0.002	1	83448		2192276		2192276			118443	1916588	
26	0.003	0	92085		2537785		..			87226	2442328	
27	0.004	0	39506		434593		..			87226	533509	
28	0.005	0	72384		1749713		..			87226	1848629	
29	0.006	0	85538		2275875		..			87226	2374791	
30	0.007	0	63936		1411823		..			87226	1510739	

Figure 6.

4. Strategic use of information

The last paragraph of the Scotia case raises the question of whether Scotia should be willing to pay some additional costs of about $100,000, to be able to base its production decision on better information about the future weather pattern that will prevail next winter. In effect, our problem is to estimate how much information about the weather pattern would be worth to Scotia. Figure 6 shows a spreadsheet to calculate the expected value of this information for Scotia, using the criterion of expected profit maximization (risk neutrality). Figure 6 was constructed from the spreadsheet shown in Figure 3, and the tables of simulated weather and demand data are the same in Figures 3 and 6.

To investigate the question of what would be optimal if Scotia knew that the weather pattern would be cold, the value 1 was entered into cell G21 of Figure 6. Then to select the profits from the D column in the cases where cold weather happens, the formula

=IF(B23=$G$21,E23,"..")

was entered into cell G23, and G23 was copied in cells G23:G1023. The result is that the G23:G1023 range selects all the profits from the E column in rows where the weather pattern is cold, but the profits are omitted (and their place is taken by the text "..") in rows where the weather pattern is normal. Excel's AVERAGE and STDEV functions (and Simtools's CE function) ignore nonnumerical data. So the formula

=AVERAGE(G23:G1023)

in cell G18 returns our estimate of Scotia's conditionally expected profit when it uses the production quantity in cell D18 and the weather pattern is cold. By changing the value of cell D18, it can be shown in this spreadsheet that an order quantity of approximately 118,000 would maximize Scotia's conditionally expected profit given cold weather, as computed in cell G18.

Next we can change the value of cell G21 to 0. The result is that the G23:G1023 range instead selects all the profits from the E column in rows where the weather pattern is normal (not cold), and the profits will be omitted in rows where the weather pattern is cold. So with G21 equal to 0, the average reported in cell G18 becomes an estimate of the conditionally expected profit given normal weather, when the production quantity is as in cell D18. Adjusting the value of cell D18 with cell G21 equal to 0, we can find that an order quantity of approximately 87,000 would maximize the conditionally expected profit given normal weather, as computed in cell G18.

The I,J,K columns of the spreadsheet show how to do this analysis in one step, using strategic analysis. Decision analysts use the word <u>strategy</u> to refer to any plan that specifies what decisions would be made as a function of any information that the decision-maker may get now and in the future. So suppose that Scotia's managers anticipate getting a perfect forecast of the winter weather pattern before they choose the production quantity, but they do not yet have the forecast in hand. In this situation, a strategy for Scotia would be any rule that specifies how much to produce if the weather forecast is normal and how much to produce if the forecast is cold.

The range I18:J19 in Figure 6 is used to display the strategy being considered. Cells I18 and I19 contain the numbers 0 and 1 respectively, to represent the cases of normal and cold forecasts. The intended order quantities in these two cases are listed in cells J18 and J19 respectively. For example, entering the value 90000 into cell J18 and 100000 into cell J19 would represent the strategy of planning to order 90,000 snowboards if the forecast is normal but 100,000 snowboards if the forecast is cold.

The formula

=VLOOKUP(B23,$I$18:$J$19,2,0)

has been entered into cell J23 in Figure 6. This formula tells Excel to look for the value of B23 in the leftmost (I) column of the range I18:J19, and then return the value found in column 2 within this range (the J column). Including 0 as the optional fourth parameter of the VLOOKUP function tells it to look for an exact match for the B23 value in the I18:I19 range. Then this formula in cell J23 has been copied to J23:J1023. The result is that each cell in J23:J1023 displays the production quantities that would be implemented under our strategy for the weather pattern given in the corresponding cell in B23:B1023. Then the profits that would result from implementing this strategy are computed in the K column by entering the formula

=($B$12−$B$10)*J23+($B$11−$B$12)*MIN(J23,C23)

into cell K23 and then copying cell K23 to K23:K1023.

Scotia's expected profit under this I18:J19 strategy is computed in cell K18 by the formula

=AVERAGE(K23:K1023)

Now we can ask Solver to maximize the target cell K18 by changing the strategic plan's production quantities in cells J18:J19. The result is that Solver will find the optimal strategy for using this forecast information to maximize Scotia's expected profit. The resulting values returned by Solver are actually shown in cells J18:J19 of Figure 6. That is, the optimal strategy found by Solver is to produce 87,226 snowboards if the forecast is normal and to produce 118,443 snowboards if the forecast is cold. In fact, these two conditional production quantities are the same as the quantities that would maximize the conditionally expected profit in G18 given each of these two weather patterns in G21.

Figure 6 also shows in cell D18 the optimal production quantity for maximizing the expected profit in cell E18 when Scotia does not get any weather forecast, as we saw in Figure 3 with the same simulation data. So the difference between cell E18 and cell K18, which is computed in cell K15, is the amount that Scotia's expected profit would be increased by allowing Scotia to use such a perfect weather forecast in its production planning. This increase in expected profit is called the expected value of information. So according to our simulation analysis, the expected value of weather information to Scotia would be approximately $45,000. This expected value of information is the most that Scotia should be willing to pay to get a perfect forecast of the winter weather pattern before choosing its production quantity. Thus, the estimated costs of $100,000 to delay production until forecast information is available are greater than the expected benefits of getting the information.

The analysis in Figure 6 could be easily adapted to use the criterion of certainty equivalent maximization instead of expected profit maximization. To apply the risk tolerance constant $1,216,303, we could simple change the estimated expected profit in K18 by the certainty equivalent formula

=CE(K23:K1023, 1216303)

and ask Solver to maximize this certainty equivalent by changing the production quantities in J18:J19. With the simulation in this spreadsheet, Solver found an opitimal strategy of producing 76,739 snowboards if the forecast is normal and producing 99,391 snowboards if the forecast is cold. The estimated certainty equivalent of the profit distribution with this risk-averse optimal strategy was found to be $1,764,277. Recall (from the discussion of Figure 3 at the end of

Section 3) that the highest certainty equivalent that could be achieved without forecast information (using the same simulation data) was found to be $1,733,204 (achieved with the production quantity 82,524). The value of weather-forecast information for Scotia with constant risk tolerance can then be estimated by the difference between these two CE values

$$\$1764277 - 1733204 = \$31,073.$$

So when risk aversion is taken into account, we still find that the production delay is not worthwhile for Scotia.

Problems for Chapter 4.

<u>1.</u> Our local heroes are going to the Rose Bowl next month, and we have a licence to sell official "Rose Bowl" sweatshirts. The sweatshirts cost $10 each to produce, and we will sell them before the Rose Bowl at a price of $25. At this price, the anticipated demand for our sweatshirts before the Rose Bowl should be a Normal random variable with mean 9000 and standard deviation 2000. After the Rose Bowl, things depend on whether our local heroes win or not. If they win, then we will continue to sell the sweatshirts for $25, and demand in the month after the game will be a Normal random variable with mean 6000 and standard deviation 2000. If they lose, then we will cut the price to $12.50 and demand at that price should be a Lognormal random variable with mean 2000 and standard deviation 1000. The probability of our local heroes winning the Rose Bowl game is 0.4. Undemanded sweatshirts will be discarded.
(a) What production quantity would maximize our expected profit?
(b) How would our optimal quantity change if we want to maximize our certainty equivalent, when we evaluate risky profits with a risk tolerance of $50,000?
(c) Assuming that we want to maximize our expected profit (as in part a), what would be the expected value of perfect information about whether our local heroes will win the Rose Bowl?

<u>2.</u> A decision-maker subjectively assessed the following certainty equivalents:
(i) "a lottery that pays either $10,000 or $0, each with probability 1/2, would be worth $3800 (for sure) to me."
(ii) "a lottery that pays either $5000 or $0, each with probability 1/2, would be worth $2200 to me."
(iii) "a lottery that pays either $1000 or $0, each with probability 1/2, would be worth $420 to me."
(a) If he has constant risk tolerance, then what risk tolerance indexes would be implied by each of these three statements?
(b) One of these statements implies a risk-tolerance index that is substantially different from the other two. If the decision-maker reassessed the certainty equivalent for the lottery described in this statement, what certainty equivalent would correspond to the same level of risk tolerance that he expressed in his other two statements?

3. A movie studio is about to release a new film. This studio has a regular policy of spending $5 million on advertising in local media to promote any new film that they release.
But for some films that are considered to have strong potential, the studio may also invest in a special marketing campaign on national television to increase interest in the film.
The special marketing campaign on national television would cost an extra $10 million (thus increasing the cost of promoting the film to a total of $15 million).

The marketing staff at the studio has tried to forecast this new film's potential revenue.
The experts at the studio believe that, if the studio follows its regular marketing policy (without any special national television campaign), the studio's revenue from this film has 50% probability of being greater than $18 million, 25% probability of being greater than $30 million, and 25% probability of being less than $12 million.
Let us assume that this revenue is drawn from a Generalized-Lognormal distribution.
The marketing experts also believe that, whatever the revenue would be under the regular marketing policy, having a special marketing campaign on national television would increase revenue by 50%.

A more complicated strategy has been suggested for this film: to first release it only in the New York metropolitan area (with only the regular local advertising), and then make a decision about national marketing based on the results of the first week in New York.
In previous films that were distributed with the regular policy, the ratio of the first week's revenue in New York divided by the total national revenue has had mean 0.02 and standard deviation 0.01.
So let us assume that the first week's revenue in New York is a random fraction, drawn from a Lognormal distribution with mean 0.02 and standard deviation 0.01, multiplied by the total national revenue that the film would earn with regular the marketing policy.

(a) Make a simulation model to analyze the studio's possible profit under these alternative strategies. Then answer parts (b) and (c) based on a table of output from at least 400 simulations of your model.

(b) Consider first the two simple alternatives (regardless of the NY results) of (b1) distributing the film with regular marketing, and (b2) distributing the film with a special marketing campaign on national television. For each of these two alternatives, estimate the mean and standard deviation of the studio's profit from distributing the film, and show the cumulative risk profile of the studio's profit from distributing the film. (Define profit as revenue minus marketing costs.)

(c) Now consider more complicated strategies of the following form:

> "first release the film only in the New York metropolitan area, then spend an extra $10 million for a special marketing campaign on national television if the first week's revenue in New York is greater than X, but otherwise (if less than X) continue with regular marketing only."

Consider different values of X from $0 up to $1 million at intervals of $0.1 million (that is, consider X = $0, X = $0.1 million, X = $0.2 millions, X = $0.3 million, ..., X = $1.0 million).
Make a table and a chart showing, for each of these strategies, the expected profit and the profit levels that have 10% and 90% cumulative probabilities for the studio from this film.
If the studio wants to maximize its expected profit, which of these strategies should it use?